

お客様各位

カタログ等資料中の旧社名の扱いについて

2010年4月1日を以ってNECエレクトロニクス株式会社及び株式会社ルネサステクノロジが合併し、両社の全ての事業が当社に承継されております。従いまして、本資料中には旧社名での表記が残っておりますが、当社の資料として有効ですので、ご理解の程宜しくお願ひ申し上げます。

ルネサスエレクトロニクス ホームページ (<http://www.renesas.com>)

2010年4月1日
ルネサスエレクトロニクス株式会社

【発行】ルネサスエレクトロニクス株式会社 (<http://www.renesas.com>)

【問い合わせ先】<http://japan.renesas.com/inquiry>

ご注意書き

1. 本資料に記載されている内容は本資料発行時点のものであり、予告なく変更することがあります。当社製品のご購入およびご使用にあたりましては、事前に当社営業窓口で最新の情報をご確認いただきますとともに、当社ホームページなどを通じて公開される情報に常にご注意ください。
2. 本資料に記載された当社製品および技術情報の使用に関連し発生した第三者の特許権、著作権その他の知的財産権の侵害等に関し、当社は、一切その責任を負いません。当社は、本資料に基づき当社または第三者の特許権、著作権その他の知的財産権を何ら許諾するものではありません。
3. 当社製品を改造、改変、複製等しないでください。
4. 本資料に記載された回路、ソフトウェアおよびこれらに関連する情報は、半導体製品の動作例、応用例を説明するものです。お客様の機器の設計において、回路、ソフトウェアおよびこれらに関連する情報を使用する場合には、お客様の責任において行ってください。これらの使用に起因しお客様または第三者に生じた損害に関し、当社は、一切その責任を負いません。
5. 輸出に際しては、「外国為替及び外国貿易法」その他輸出関連法令を遵守し、かかる法令の定めるところにより必要な手続を行ってください。本資料に記載されている当社製品および技術を大量破壊兵器の開発等の目的、軍事利用の目的その他軍事事務の目的で使用しないでください。また、当社製品および技術を国内外の法令および規則により製造・使用・販売を禁止されている機器に使用することができません。
6. 本資料に記載されている情報は、正確を期すため慎重に作成したのですが、誤りがないことを保証するものではありません。万一、本資料に記載されている情報の誤りに起因する損害がお客様に生じた場合においても、当社は、一切その責任を負いません。
7. 当社は、当社製品の品質水準を「標準水準」、「高品質水準」および「特定水準」に分類しております。また、各品質水準は、以下に示す用途に製品が使われることを意図しておりますので、当社製品の品質水準をご確認ください。お客様は、当社の文書による事前の承諾を得ることなく、「特定水準」に分類された用途に当社製品を使用することができません。また、お客様は、当社の文書による事前の承諾を得ることなく、意図されていない用途に当社製品を使用することができません。当社の文書による事前の承諾を得ることなく、「特定水準」に分類された用途または意図されていない用途に当社製品を使用したことによりお客様または第三者に生じた損害等に関し、当社は、一切その責任を負いません。なお、当社製品のデータ・シート、データ・ブック等の資料で特に品質水準の表示がない場合は、標準水準製品であることを表します。
標準水準： コンピュータ、OA 機器、通信機器、計測機器、AV 機器、家電、工作機械、パーソナル機器、産業用ロボット
高品質水準： 輸送機器（自動車、電車、船舶等）、交通用信号機器、防災・防犯装置、各種安全装置、生命維持を目的として設計されていない医療機器（厚生労働省定義の管理医療機器に相当）
特定水準： 航空機器、航空宇宙機器、海中継機器、原子力制御システム、生命維持のための医療機器（生命維持装置、人体に埋め込み使用するもの、治療行為（患部切り出し等）を行うもの、その他直接人命に影響を与えるもの）（厚生労働省定義の高度管理医療機器に相当）またはシステム等
8. 本資料に記載された当社製品のご使用につき、特に、最大定格、動作電源電圧範囲、放熱特性、実装条件その他諸条件につきましては、当社保証範囲内でご使用ください。当社保証範囲を超えて当社製品をご使用された場合の故障および事故につきましては、当社は、一切その責任を負いません。
9. 当社は、当社製品の品質および信頼性の向上に努めておりますが、半導体製品はある確率で故障が発生したり、使用条件によっては誤動作したりする場合があります。また、当社製品は耐放射線設計については行っておりません。当社製品の故障または誤動作が生じた場合も、人身事故、火災事故、社会的損害などを生じさせないようお客様の責任において冗長設計、延焼対策設計、誤動作防止設計等の安全設計およびエージング処理等、機器またはシステムとしての出荷保証をお願いいたします。特に、マイコンソフトウェアは、単独での検証は困難なため、お客様が製造された最終の機器・システムとしての安全検証をお願いいたします。
10. 当社製品の環境適合性等、詳細につきましては製品個別に必ず当社営業窓口までお問合せください。ご使用に際しては、特定の物質の含有・使用を規制する RoHS 指令等、適用される環境関連法令を十分調査のうえ、かかる法令に適合するようご使用ください。お客様がかかる法令を遵守しないことにより生じた損害に関し、当社は、一切その責任を負いません。
11. 本資料の全部または一部を当社の文書による事前の承諾を得ることなく転載または複製することを固くお断りいたします。
12. 本資料に関する詳細についてのお問い合わせその他お気付きの点等がございましたら当社営業窓口までご照会ください。

注 1. 本資料において使用されている「当社」とは、ルネサスエレクトロニクス株式会社およびルネサスエレクトロニクス株式会社とその総株主の議決権の過半数を直接または間接に保有する会社をいいます。

注 2. 本資料において使用されている「当社製品」とは、注 1 において定義された当社の開発、製造製品をいいます。

H8/3069F グループ

0.18 μ m プロセス F-ZTAT マイコンオンボード書き込み

目次

1.	0.18 μ m プロセス F-ZTAT の特徴	2
2.	0.35 μ m プロセス F-ZTAT との相違点	3
3.	消去／書き込み手続きプログラム概要	5
4.	オンボードプログラミングモード (H8/3069F の場合)	7
4.1	H8/3069F のフラッシュの特徴	7
4.2	ブートモード概要	9
4.3	ユーザプログラムモード概要	11
4.4	ユーザブートモード概要	14
5.	フラッシュ書き込み手順～H8/3069F の場合	18
5.1	ブートモード書き込み手順	18
5.1.1	ビットレート問い合わせステータス	18
5.1.2	問い合わせ選択ステータス	18
5.1.3	消去／書き込みステータス	19
5.2	ユーザプログラムモード書き込み手順	21
5.3	ユーザブートモード書き込み手順	27
6.	ユーザ作成プログラム例	31
7.	H8/3069F 消去／書き込み用レジスタ・パラメーター一覧	38

1. 0.18 μ m プロセス F-ZTAT の特徴

F-ZTAT*マイコンは、オンボード（機器組み込み状態）でプログラムの消去／書き込みが可能なフラッシュメモリ内蔵マイコンです。オンボードでプログラム書き換えが可能なので、市場出荷後のソフトウェアのバージョンアップ、組み込み機器ごとのパラメータの最適化、メンテナンスなどを容易に行うことができます。

0.18 μ m プロセス F-ZTAT には、専用の消去／書き込みプログラムが内蔵されています。

消去／書き込みを行うときは、それらの内蔵モジュールを内蔵 RAM にダウンロード後、所定の引数パラメータをユーザシステムにあわせ設定し、呼び出すだけで、消去／書き込みが自動で行えるようになりました。

消去は、消去ブロック単位で行い、一度の消去プログラム呼び出しで 1 ブロックの消去ができます。

書き込みは、128 バイト単位で行い、一度の書き込みプログラム呼び出しで、128 バイトの書き込みができます。

【注】 * F-ZTAT (Flexible Zero Turn Around Time) は (株) ルネサステクノロジーの商標です。

2. 0.35 μm プロセス F-ZTAT との相違点

図2.1に0.35 μmプロセスF-ZTATの消去／書き込み、図2.2に0.18 μmプロセスF-ZTATの消去／書き込み方法を示します。

0.35 μmプロセスF-ZTATでは、消去／書き込みを行う場合、消去プログラム、書き込みプログラム、それらを制御する手続きプログラムのすべてを作成しなければなりません。

それに対し、0.18 μmプロセスF-ZTATでは、消去プログラム、書き込みプログラムが、マイコンに内蔵されているため、手続きプログラムを作成すればよいことになりました。

消去／書き込みの際には、手続きプログラムをRAM上に展開し実行します。手続きプログラムの処理内で、内蔵プログラム(初期化、消去、書き込み)をRAM上にダウンロードし、それらをサブルーチン呼び出して、消去／書き込みを行います。




図 2.1 0.35 μm プロセス F-ZTAT の消去／書き込み




図 2.2 0.18 μ m プロセス F-ZTAT の消去/書き込み

3. 消去／書き込み手続きプログラム概要

0.18 μmプロセスF-ZTATの消去／書き込み手続きプログラムは、図3.1に記載するシーケンスで作成します。

この図において、初期化の実行、消去実行、書き込み実行の各シーケンスはダウンロードした内蔵プログラムをサブルーチン呼び出しすることで実行します。

内蔵プログラムをダウンロードするときは、消去／書き込み、どちらの処理を実行するかにより、消去プログラムか、書き込みプログラムかを選択します。ただし、一度のダウンロードで、消去処理、書き込み処理をともに実行することはできません。

フラッシュを消去して書き込みを行う場合は、「消去プログラムのダウンロード～消去完了」、「書き込みプログラムのダウンロード～書き込み完了」と処理を2回にわけて行います。




図 3.1 消去／書き込みプログラム概要

消去／書き込み手続きプログラムで実行される処理の概要を記述します。消去／書き込み手続きプログラムを作成する場合、これらの処理を組み込みます。

(1) ダウンロードのパラメータ指定

- ・ ダウンロード先の内蔵RAMアドレスの選択 (FTDAR)
- ・ ダウンロードプログラム (消去or書き込み) の選択 (FPCSのPPVSビット or FECSのEPVBビット)

(2) RAM 上への内蔵プログラムダウンロード

- フラッシュキーコードレジスタ(FKEY)にH'A5を設定
 - フラッシュコードコントロール・ステータスレジスタ (FCCS) を設定しダウンロード
- ダウンロード実行後、ダウンロードパス・フェイルリザルトパラメータ (DPFR) を参照し、エラーチェック

(3) 初期化の実行

ダウンロードした初期化プログラムを実行 (サブルーチン呼び出し) します。初期化プログラムのエントリアドレスは、(FTDARで設定したアドレス+32) 番地です。

初期化実行の際に設定するレジスタ、パラメータは、次のとおりです。

- CPUの動作周波数の設定 (FPEFEQ)
 - ユーザブランチアドレスの設定 (FUBRA) (未サポートLSIあり)
- 初期化実行後、フラッシュパス/フェイルリザルトパラメータ (FPFR) を参照し、エラーチェック

(4) 消去/書き込みの実行

ダウンロードした消去プログラムまたは、書き込みプログラムを実行 (サブルーチン呼び出し) することによって消去/書き込みは行われます。消去/書き込みプログラムのエントリアドレスは、(FTDARで設定したアドレス+16) 番地です。

消去/書き込み実行の際に設定するレジスタ、パラメータは、次のとおりです。

1. 消去/書き込み共通のレジスタ設定
 - フラッシュキーコードレジスタ (FKEY) に H'5A を設定
 2. 書き込み時のパラメータ設定
 - フラッシュ ROM 書き込み先の先頭アドレス (FMPAR)
 - 書き込みデータ格納領域の先頭アドレス (FMPDR)
 3. 消去時のパラメータ設定
 - 消去ブロック番号 (FEBS)
- 消去/書き込み実行後、フラッシュパス/フェイルリザルトパラメータ (FPFR) を参照し、エラーチェック

4. オンボードプログラミングモード (H8/3069F の場合)

0.18 μmプロセスF-ZTATマイコンの消去書き込み方法を説明します。ここでは、0.18 μmプロセスF-ZTATマイコンであるH8/3069Fを取り上げ、その特徴とマイコンが待つ3種類のオンボードプログラミングモードについて説明します。

4.1 H8/3069F のフラッシュの特徴

まず初めに、0.18 μmプロセスF-ZTATマイコンH8/3069Fに関する特徴を述べます。

ここでは、H8/3069Fが持つ2種類のフラッシュメモリと、3種類のオンボードプログラミングモードについて記述します。

(1) 2種類のフラッシュメモリ内蔵

0.18 μmプロセスのF-ZTATは、フラッシュメモリの構成は、2種類のメモリ空間からなっています。これらはメモリマットと呼ばれ、512kBのユーザマット、8kBのユーザブートマットから成り立ちます (図4.1参照)。

使用用途としては、ユーザマットは、通常、動作するアプリケーションプログラムを格納します。

ユーザブートマットは、後述するユーザブートモードで使用し、フラッシュ消去/書き込み手続きプログラムを格納します。

ユーザマット、ユーザブートマットの先頭アドレスは、同一アドレスであるため、2つのマット間で実行したり、データアクセスがまたがる場合には必ずマットの切り替えを行わなければなりません。ユーザマットの書き込みは128バイト単位、消去は図4.2に示すブロック単位で行います。




図 4.1 フラッシュメモリ構造




図 4.2 消去ブロック構造

(2) 3種類のオンボードプログラミングモード

フラッシュの消去／書き込みを行う方法には、オンボードプログラミングモードとして、次の3種類のモードが用意されています。これらのモードを選択するには、表4.1に示すとおり端子の設定を行うことで、任意のモードでの消去／書き込みを行うことができます。

- ブートモード
内蔵SCIインタフェースを使用します。ホストとF-ZTATマイコンに内蔵されたブートプログラムとの間でコマンド、データのやり取りを行い、ユーザマット、ユーザブートマットの消去／書き込みを行います。
- ユーザプログラムモード
任意のインタフェースでの、ユーザマットの書き換えが可能です。
- ユーザブートモード
ユーザ作成のユーザブートプログラムにより、任意のインタフェースでの、ユーザマットの書き換えが可能です。このモードは、ユーザブートマットから起動されるため、ユーザマットへの消去／書き込み処理では、マットの切り替えを行う必要があります。

表 4.1 モード別端子設定

端子	モード				
	リセット 状態	書き換えモード			アプリケーション 実行時
		ユーザプログラム モード	ユーザブート モード	ブートモード	
RES	0	1	1	1	1
FEW	0/1	1	1	1	0
MD0	0/1	1	1	1	0/1
MD1	0/1	0/1	0/1	0/1	0/1
MD2	0/1	1	0	0	0/1
NMI	0/1	0/1	0	1	0/1

4.2 ブートモード概要

ブートモードは、マイコン内蔵のSCIを利用してホストから制御コマンドや書き込みデータを送受信する方式で、ユーザマットやユーザブートマットへの消去/書き込みを実行します。

ブートモード時のシステム構成を図4.3に示します。

表4.1に示したブートモードに各々端子設定し、リセットスタートすることでブートプログラムが起動します。ブートプログラムはSCIビットレートの自動調整後、ホストから送信される制御コマンドを受け、そのコマンドに対するレスポンスをホストに返信します。そして、受信コマンドが書き込みや消去などの処理を伴うコマンドであれば、それらのコマンド処理を実行していきます。

ホスト上には制御コマンドを送信するためのツールと書き込みデータを準備しておく必要があります。




図 4.3 ブートモード時システム構成

ホストとブートプログラムのコマンドのやり取りを大きく分けると、次の3状態に分けることができます。1から3の順で遷移していきます。

1. ビットレート合せ込みステータス
2. 問い合わせステータス
3. 消去／書き込みステータス

これらのステータスの遷移については、図4.4を参照してください。




図 4.4 ブートモードステータス遷移

4.3 ユーザプログラムモード概要

ユーザプログラムモードでは、ユーザマツトの消去／書き込みが可能です。マイコンに内蔵されている消去／書き込みプログラムをダウンロードして使用します。内蔵プログラムのダウンロードの要求、消去／書き込みの制御、結果の判定など行う手続きプログラムをユーザマツト上に作成します。消去／書き込みを行うときには作成した手続きプログラムをRAM上に展開させ実行させます。ユーザプログラムモードの処理の流れを図4.5に示します。

(1) 手続きプログラムのRAM展開

ユーザマツト上の手続きプログラムをRAM上に展開します。以降の制御は、すべてRAM上に展開した手続きプログラムにて行います。




図 4.5(A) 手続きプログラムのRAM 展開

(2) 内蔵プログラムのダウンロード

消去時には消去プログラムを、書き込み時には書き込みプログラムをRAM上にダウンロードします。

同時に、初期化プログラムがダウンロードされます。




図 4.5(B) 内蔵プログラムダウンロード




図 4.5(C) ユーザマット書き込み

(3) ユーザマットの書き込み

RAM上にダウンロードした初期化プログラムをサブルーチン呼び出し、書き込みに必要な初期化処理を実行します。

↓ 初期化終了

RAM上にダウンロードした書き込みプログラムをサブルーチン呼び出しすることで、ユーザマットの書き込みを行います。このとき、ユーザマットへの書き込みは128バイト単位で行われるので、書き込みプログラムの呼び出しは128バイトごとに行い、すべての書き込みデータ(アプリケーションプログラム)を書き込み完了するまで続けます。

(4) ユーザマットの消去

RAM上にダウンロードした初期化プログラムをサブルーチン呼び出し、消去に必要な初期化処理を実行します。

↓ 初期化終了

RAM上にダウンロードした消去プログラムをサブルーチン呼び出しすることで、ユーザマットの消去を行います。このとき、ユーザマットの消去はブロック単位で行われるので、消去プログラムの呼び出しは1ブロックごとに行い、すべての消去対象ブロックが消去されるまで続けます。




図 4.5(D) ユーザマット消去

4.4 ユーザブートモード概要

ユーザブートモードでは、内蔵SCI以外の任意のインタフェースのブートモードを実現できます。

ユーザブートモードの起動方法は、表 4.1 の端子設定を参照してください。ユーザブートモードでの消去／書き込みでは、手続きプログラムはユーザブートマット上に作成しておきます。また、ユーザブートモード起動時はユーザブートマットが選択されているため、ユーザマットの消去／書き込みを行う際には、ユーザマットへの切り替えを行う必要があります。

ユーザプログラムモードの処理の流れを図4.6に示します。

(1) 手続きプログラムの RAM 展開

ユーザブートマット上の手続きプログラムをRAM上に展開します。以降の制御は、すべてRAM上に展開した手続きプログラムにて行います。




図 4.6(A) 手続きプログラムのRAM 展開

(2) 内蔵プログラムのダウンロード

消去時には消去プログラムを、書き込み時には書き込みプログラムをRAM上にダウンロードします。同時に、初期化プログラムがダウンロードされます。




図 4.6(B) 内蔵プログラムのダウンロード

(3) ユーザマットの書き込み

RAM上にダウンロードした初期化プログラムをサブルーチン呼び出し、書き込みに必要な初期化処理を実行します。

↓ 初期化終了

RAM上にダウンロードした書き込みプログラムをサブルーチン呼び出しすることで、ユーザマットの書き込みを行います。このとき、ユーザマットへの書き込みは128バイト単位で行われるので、書き込みプログラムの呼び出しは128バイトごとに行い、すべての書き込みデータ(アプリケーションプログラム)を書き込み完了するまで続けます。

【注】 ユーザマット書き込み時には、ユーザブートマットからユーザマットへの切り替えが必要です。




図 4-6(C) ユーザマット書き込み

(4) ユーザマットの消去

RAM上にダウンロードした初期化プログラムをサブルーチン呼び出し、消去に必要な初期化処理を実行します。

↓ 初期化終了

RAM上にダウンロードした消去プログラムをサブルーチン呼び出しすることで、ユーザマットの消去を行います。このとき、ユーザマットの消去はブロック単位で行われるので、消去プログラムの呼び出しは1ブロックごとに行い、すべての消去対象ブロックが消去されるまで続けます。

【注】 ユーザマット消去時には、ユーザブートマットからユーザマットへの切り替えが必要です。




図 4.6(D) ユーザマツト消去

5. フラッシュ書き込み手順～H8/3069F の場合

H8/3069Fを例にして、0.18 μ m プロセスF-ZTATのフラッシュ書き込み手順を3種類のオンボードプログラミングモードごとに説明します。

5.1 ブートモード書き込み手順

ブートモードでリセットスタート後、ブートプログラムは次の3ステータスを持ちます。各ステータスの概要については、4.2のブートモード概要を参照してください。

各ステータスにおいて、ホストとブートプログラムのインタフェースについて述べます。

5.1.1 ビットレート問い合わせステータス

ビットレート問い合わせステータスでは、ホストからSCI調歩同期式でH'00を送信し続けます。ブートプログラムでは、送信データのLOW期間を測定します。LOW期間の測定から、ホストの送信するビットレートを計算し、合わせ込み終了ステータス (H'00) をホストへ送信します。

ホストは、H'00を正常に受信したことを確認し、ブートプログラムに対し、H'55を送信します。




図 5-1 ビットレート問い合わせシーケンス

5.1.2 問い合わせ選択ステータス

問い合わせ選択ステータスでは、ブートプログラムはホストからの問い合わせコマンドに対してフラッシュROMの情報を応答し、選択コマンドに対しては、デバイス、クロックモード、ビットレートを選択します。

ホストは送信した問い合わせコマンドに対するレスポンスのチェックを行います。入力周波数、通倍比、動作周波数、ビットレートに対するチェック方法は、ハードウェアマニュアルを参照してください。

消去／書き込みステータスに遷移するには、必ず「デバイス選択(H'20)」、「クロックモード選択(H'11)」、「新ビットレート選択(H'3F)」の順に発行した後、ホストから「書き込み消去ステータス遷移 (H'40)」を発行してください。

ブートプログラムは、自動的にユーザマット、ユーザブートマットのデータを消去し、消去が終わると、ACK (H'06) を応答し、消去／書き込みステータスに遷移します。

問い合わせ／選択コマンドについては表5.1を参照ください。各コマンドの詳細は、ハードウェアマニュアルを参照してください。

表 5.1 問い合わせ／選択コマンド一覧

コマンド	コマンド名	機能
H'20	サポートデバイス問い合わせ	デバイスコードと F-ZTAT 品名の問い合わせ
H'10	デバイス選択	デバイスコードの選択
H'21	クロックモード問い合わせ	クロックモード数とそれぞれの値の問い合わせ
H'11	クロックモード選択	選択されているクロックモードの通知
H'22	逡倍比問い合わせ	逡倍比または分周比の種類数とそれぞれの個数とその値の問い合わせ
H'23	動作周波数問い合わせ	メインクロックとペリフェラルクロックの最大値／最小値の問い合わせ
H'24	ユーザブートマット情報問い合わせ	ユーザブートマットの個数とそれぞれの先頭／最終アドレスの問い合わせ
H'25	ユーザマット情報問い合わせ	ユーザマットの個数とそれぞれの先頭／最終アドレスの問い合わせ
H'26	消去ブロック情報問い合わせ	消去ブロック数とそれぞれの先頭／最終アドレスの問い合わせ
H'27	書き込みサイズ問い合わせ	書き込み時のデータ長の問い合わせ
H'3F	新ビットレート問い合わせ	新ビットレートの選択
H'40	書き込み消去ステータス遷移	ユーザブートマット、ユーザマットを消去し、書き込み消去ステータスに遷移
H'4F	ブートプログラムステータス問い合わせ	ブート処理状態の問い合わせ

5.1.3 消去／書き込みステータス

- 書き込み処理
書き込みを行うには、最初にホストから書き込み選択コマンドを発行して、書き込み方式と書き込みマットを選択します。次に「128バイト書き込みコマンド (H'40)」を発行し、データを書き込みます。
- 消去処理
消去を行うには、最初にホスト側から消去コマンドを発行します。次に「ブロック消去コマンド (H'58)」を発行して、指定したブロックを消去します。
消去／書き込みコマンドについては表5.2を参照ください。各コマンドの詳細は、ハードウェアマニュアルを参照してください。

表 5.2 書き込み消去コマンド一覧

コマンド	コマンド名	機能
H'42	ユーザブートマット書き込み選択	ユーザブートマット書き込みプログラムの転送
H'43	ユーザマット書き込み選択	ユーザマット書き込みプログラムの転送
H'50	128 バイト書き込み	128 バイト書き込み
H'48	消去選択	消去プログラムの転送
H'58	ブロック消去	ブロックデータの消去
H'52	メモリリード	メモリの読み出し
H'4A	ユーザブートマットのサムチェック	ユーザブートマットのサムチェック
H'4B	ユーザマットのサムチェック	ユーザマットのサムチェック
H'4C	ユーザブートマットのブランクチェック	ユーザブートマットのブランクチェック
H'4D	ユーザマットのブランクチェック	ユーザマットのブランクチェック
H'4F	ブートプログラムステータス問い合わせ	ブート処理状態の問い合わせ

5.2 ユーザプログラムモード書き込み手順

ユーザプログラムモードでは、ユーザでユーザマットの消去／書き込みの制御を行うプログラムを作成します。消去／書き込みは、RAM上で動作させます。

ユーザが作成する必要があるのは次のプログラムです。

1. RAM上で動作させる消去／書き込み手続きプログラム
2. 1.のプログラムをRAM上に展開するプログラム

128バイト単位の書き込みプログラム、1ブロック単位の消去プログラムは、マイコンに内蔵されています。ユーザはそれらの内蔵プログラムをRAM上へのダウンロードする処理、そして、書き込みプログラムまたは、消去プログラムの呼び出しを行う手続きプログラムを作成する必要があります。初期化プログラムは、消去／書き込みプログラムのダウンロード時に同時にダウンロードされます。

書き込みを行う前には、書き込むエリアが消去されている必要があります。消去されていない場合には、必ず消去処理実行後、書き込み処理を行ってください。

図5.2にフラッシュメモリ(ユーザマット)の消去／書き込みの概略図を示します。




図 5.2 ユーザプログラムモード書き込み／消去処理概略

消去／書き込み手続きプログラムで設定を行うレジスタについて説明します。図5.2内の右枠に示した4処理 [(1) ダウンロードのパラメータ指定、(2) ダウンロードの実行、(3) 初期化の実行、(4) 消去／書き込みの実行] 別に記述します。

(1) ダウンロードのパラメータ指定

消去／書き込みを行うには、まず、内蔵プログラムなどをRAMにダウンロードします。ダウンロード後のRAMの状態は図5.3のような構成になります。ダウンロード実施前に設定が必要なレジスタは次のとおりです。

- ダウンロード先の内蔵RAMアドレス指定
フラッシュトランスファディステーションレジスタ (FTDAR)
 内蔵プログラムやその関連情報をダウンロードする時に、ダウンロード先の内蔵RAMのアドレスを指定します。このレジスタの設定でダウンロード先の先頭アドレスを幾通りかの決められたアドレスの中から選択することができます。
- ダウンロードするプログラム(消去プログラム、書き込みプログラム)の選択
フラッシュプログラムコードセレクトレジスタ (FPCS)
 内蔵書き込みプログラムのダウンロードをする場合に使用します。書き込み時には、本レジスタのPPVS(ビット0)を“1”にセットします。
フラッシュイレースコードセレクトレジスタ (FECS)
 内蔵消去プログラムのダウンロードをする場合に使用します。消去時には、本レジスタのEPVB(ビット0)を“1”にセットします。




図 5-3 内蔵プログラムダウンロード実行後の RAM マップ

(2) ダウンロードの実行

ダウンロード実行時には、次のレジスタを設定します。

- フラッシュキーコードレジスタ(FKEY)
 ダウンロード時には、本レジスタにH'A5をセットしてください。それ以外の値がセットされていると、ソフトウェアプロテクト状態となり、FCCSのSCOに書き込むことができません。ダウンロード後はH'00にセットし、ソフトウェアプロテクト状態としてください。
- フラッシュコードコントロール・ステータスレジスタ(FCCS)

本レジスタのSCO(ビット0)に“1”を書き込むことで、指定した内蔵RAM領域に指定されたプログラムをダウンロードします。

(3) 初期化の実行

消去／書き込みの実行前には初期化処理を行います。これはダウンロードした初期化プログラムを実行(サブルーチン呼び出し)することによって行います。初期化プログラムのエントリアドレスは、FTDARで設定したアドレス+32番地に格納されています。初期化実行時の参考プログラムを図5.4に示します。

また、初期化実行の際に設定するパラメータは次のとおりです。

- フラッシュプログラム/イレース周波数パラメータ (FPEFEQ:ER0)
 CPUの動作周波数を設定します。
 (設定例) 25MHz動作の場合は、25.00(少数第3位は四捨五入)×100=2500をER0に設定
- フラッシュユーザブランチアドレス設定パラメータ (FUBRA:ER1)
 消去／書き込み実行時のある決まった処理単位ごとに、設定したユーザプログラムを実行することができます(ただし、H8/3069Fでは未サポート)。
- フラッシュパス/フェイルリザルトパラメータ (FPFR:ROL)
 初期化結果がFPFR(ROL)に格納されます。ROLがH'00以外の場合エラーとなります。

繰り返し消去／書き込みを実行する場合、初期化処理は消去／書き込み前にそれぞれ一度実行するだけでかまいません。

```

<<初期化処理実行>> ...ENTRYはFTDARレジスタで指定したアドレス
MOV.L ENTRY+32,R0 ;初期化プログラム格納アドレス指定
JSR @R0 ;初期化プログラム実行
    
```

図 5.4 初期化プログラム実行例

(4) 消去／書き込みの実行

消去／書き込みの実行時にはダウンロードした消去プログラム／書き込みプログラムを実行(サブルーチン呼び出し)することによって行います。消去／書き込みプログラムのエントリアドレスは、FTDARで設定したアドレス+16番地に格納されています。消去／書き込み実行の参考プログラムを図5.5に示します。

ここで、ダウンロードした消去プログラムは1ブロックの消去、書き込みプログラムは128バイト単位の書き込みとなっています。したがって、複数のブロック消去、128バイトを超えた書き込みを行う場合には、繰り返し消去／書き込みプログラムを実行する必要があります。

また、消去／書き込み実行の際に設定するパラメータは次のとおりです。

- 消去／書き込み時、共通のパラメータ設定
フラッシュキーコードレジスタ(FKEY)
 消去／書き込み時には、本レジスタにH'5Aをセットしてください。それ以外の値がセットされていると、ソフトウェアプロテクト状態となり、消去／書き込みは実行されません。消去／書き込み後にはH'00にセットし、ソフトウェアプロテクト状態としてください。
フラッシュマットセレクトレジスタ (FMATS)
 ユーザマット選択/ユーザブートマットの切り替えを行うときに設定します。
 ユーザマットはH'00、ユーザブートマットはH'AAを指定します。

- 書き込み時のパラメータ設定
フラッシュマルチパースアドレスエリアパラメータ (FMPAR : ER1)
 ユーザマット書き込み先の先頭アドレスを指定します。
フラッシュマルチパースデータデスティネーションパラメータ (FMPDR : ER0)
 ユーザマットに書き込むデータが格納されている領域の先頭アドレスを設定します。
フラッシュパス/フェイルリザルトパラメータ (FPFR : R0L)
 書き込み結果がFPFR (R0L) に格納されます。R0LがH'00以外の場合エラーとなります。
- 消去時のパラメータ設定
フラッシュイレースブロックセレクトパラメータ (FEBS : ER0)
 消去するブロック番号を指定します。
フラッシュパス/フェイルリザルトパラメータ (FPFR : R0L)
 消去結果がFPFR(R0L)に格納されます。R0LがH'00以外の場合エラーとなります。

```

<<消去/書き込み処理実行>>...ENTRYはFTDARレジスタで指定したアドレス
MOV.L  ENTRY+16,R0          ;消去/書き込みプログラム格納アドレス指定
JSR    @R0                  ;消去/書き込みプログラム実行
    
```

図 5.5 消去/書き込みプログラム実行例

ユーザプログラムモードにおける書き込み手順を次に示します。図5.6に書き込み手続きプログラムのフロー図を示します。

1. FWE端子をハイレベルに設定
2. 消去/書き込み手続きプログラムを内蔵RAMに転送し、RAM上で実行
3. FTDARによりダウンロードするRAMのアドレスを指定
4. ダウンロードする内蔵プログラムを選択
5. 書き込みを行う場合、FPCSレジスタのPPVSビットを1に設定して、書き込みプログラムを選択します。
6. FKEYレジスタにH'A5の書き込み
7. FCCSレジスタのSCOビットに1を書き込みます。これによりダウンロードが実行されます。
8. SCOビット書き込み後、プロテクトのため、FKEYレジスタはH'00にクリアします。
9. ダウンロード結果のチェック
10. DFPRパラメータのSSビット、FKビット、SFビットが0であることを確認します。1が立っていた場合エラー処理を行います。
11. 初期化のためのパラメータセット
12. FPEFEQパラメータ(ER0)にCPUクロック周波数、FUBRAパラメータ (ER1) にユーザブランチ先頭アドレスを設定します(H8/3069Fではユーザブランチ未サポート)。
13. 初期化実行
14. 書き込みプログラムのダウンロード時、初期化プログラムもいっしょにダウンロードされています。初期化プログラムのエントリーポイントは、(FTDAR設定値+32)番地のアドレスに展開されています。当アドレスをサブルーチンコールし、実行させることで、初期化が行われます。
15. 初期化結果のチェック
16. FPFRパラメータのBRビット、FQビット、SFビットが0であることを確認します。1が立っていた場合エラー処理を行います。
17. すべての割り込み、CPU以外のバスマスタの動作を禁止

18. FKEYレジスタにH'5Aの書き込み
19. ユーザマットの書き込み先の先頭アドレスをFMPARパラメータ (ER1) にユーザマットに書き込むデータが格納されている領域の先頭アドレスを、FMPDRパラメータ (ER0) に設定。
20. 書き込み実行
21. (FTDAR設定値+16) 番地のアドレスに書き込みプログラムのエン트리ポイントがあります。当アドレスをサブルーチンコールし、実行させることで、128バイトの書き込みが行われます。
22. 書き込み結果のチェック
23. FPFDRパラメータのMDビット、EEビット、FKビット、WDビット、WAビット、SFビットが0であることを確認します。1が立っていた場合、エラー処理を行います。
24. 書き込み対象のデータがすべて書き込まれていなければ (13) に戻り、繰り返す。
25. FKEYレジスタをH'00にクリア




図 5.6 ユーザプログラムモード書き込み手続きプログラムフロー

ユーザプログラムモードにおける消去手順を次に示します。図5.7に消去手続きプログラムのフロー図を示します。

1. FWE端子をハイレベルに設定
2. 消去／書き込み手続きプログラムを内蔵RAMに転送し、RAM上で実行
3. FTDARによりダウンロードするRAMのアドレスを指定
4. ダウンロードする内蔵プログラムを選択
5. 消去を行う場合、FECSレジスタのEPVBビットを1に設定して、消去プログラムを選択します。
6. FKEYレジスタにH'A5の書き込み
7. FCCSレジスタのSCOビットに1を書き込みます。これによりダウンロードが実行されます。
8. SCOビット書き込み後、プロテクトのため、FKEYレジスタはH'00にクリアします。
9. ダウンロード結果のチェック
10. DPFRパラメータのSSビット、FKビット、SFビットが0であることを確認します。1が立っていた場合エラー処理を行います。
11. 初期化のためのパラメータセット
12. FPEFEQパラメータ(ER0)にCPUクロック周波数、FUBRAパラメータ(ER1)にユーザブランチ先頭アドレスを設定します(H8/3069Fではユーザブランチ未サポート)。
13. 初期化実行
14. 消去プログラムのダウンロード時、初期化プログラムもいっしょにダウンロードされています。初期化プログラムのエントリーポイントは、(FTDAR設定値+32)番地のアドレスに展開されています。当アドレスをサブルーチンコールし、実行させることで、初期化が行われます。
15. 初期化結果のチェック
16. PFRパラメータのBRビット、FQビット、SFビットが0であることを確認します。1が立っていた場合エラー処理を行います。
17. すべての割り込み、CPU以外のバスマスタの動作を禁止
18. FKEYレジスタにH'5Aの書き込み
19. FEBSパラメータの設定
20. ユーザマトの消去ブロック番号をFEBS (ER0) に設定します。
21. 消去実行
22. (FTDAR設定値+16)番地のアドレスに消去プログラムのエントリーポイントがあります。当アドレスをサブルーチンコールし、実行させることで、指定した1ブロックの消去が行われます。
23. 消去結果のチェック
24. PFRパラメータのMDビット、EEビット、FKビット、EBビット、SFビットが0であることを確認します。1が立っていた場合エラー処理を行います。
25. 消去対象のブロックがすべて消去されていなければ(13)に戻り、繰り返す。
26. FKEYレジスタをH'00にクリア




図 5.7 ユーザプログラムモード消去手続きプログラムフロー

5.3 ユーザブートモード書き込み手順

ユーザブートモードでの消去／書き込み時の処理の流れを図5-8に示します。

ユーザプログラムモードと同様、ユーザは手続きプログラムを作成する必要があります。

消去／書き込み手続きプログラムの処理において、ユーザプログラムモードとの違いは、マットの切り替えを行うか否かです。ユーザブートモードでは、リセットスタート後には、ユーザブートマットが選択されています。

ユーザマットに書き込みを行う場合には、ユーザブートマットからユーザマットへのマット切り替えが必要となり、書き込み終了後には、ユーザマットからユーザブートマットへ切り替える必要があります。

ユーザが手続きプログラムを作成する際には、このマット切り替えの処理を組み込まなくてはなりません。

ユーザブートマットからユーザマットに切り替えるタイミングは書き込み／消去開始前（FKEYレジスタにH'5A設定前）、ユーザマットからユーザブートマットに切り替えるタイミングは消去／書き込み終了後（FKEYレジスタクリア後）となります。




図 5.8 ユーザブートモード処理手順

ユーザブートモードにおける書き込み手続きプログラムのフローを図5.9に示します。ユーザプログラムモードとの違いは、網掛けになっている部分の処理が追加されます。

ユーザマットへの書き込み処理の際にマットの切り替えが必要となります。




図 5.9 ユーザーブートモード書き込み手続きプログラムフロー

ユーザーブートモードにおける消去手続きプログラムのフローを図5.10に示します。ユーザープログラムモードとの違いは、網掛けになっている部分の処理が追加されます。

ユーザーマットへの書き込み処理の際にマットの切り替えが必要となります。




図 5.10 ユーザブートモード消去手続きプログラムフロー

6. ユーザ作成プログラム例

ユーザプログラムモードで作成したプログラム例を紹介します。

このプログラム例では、定数ERASEBLKで指定したブロックに対し、消去を行い、続いてFW_STADR（ユーザマットの書き込み先頭アドレス）からFW_EDADR（ユーザマットの書き込み終了アドレス）まで、書き込みを行います。

```

;*****
;*
;* ルネサスF-ZTATマイコンH8/3069F用
;* フラッシュメモリ消去/書き込み制御プログラム (御参考)
;*
;* Ver.1.0
;*****
;*****
;* FLASHメモリ消去書き込みプログラム
;*****
;* モジュール名 ; _FLASH_W
;* -----
;* 組み込み方法
;* 1. 下記のセクションを登録してください。
;* .SECTION FCODE, CODE, ALIGN=2 (ROM エリア)
;* .SECTION FRAM, DATA, ALIGN=2 (RAM エリア)
;*
;* 2. 下記のモジュールを新規に作成してください。
;* DATAREAD128 :外部デバイスなどから128バイトのデータを読み込み
;* _FWRITE_BUFFERに格納します。
;* ただし次のようなインタフェースになっています。
;* in :なし
;* out :_FWRITE_BUFFER 読み込みデータ128バイト
;* 破壊禁止レジスタ:
;* ER5 書き込み先先頭アドレス (FLASH)
;*
;* 3. 下記のシンボルを仕様にそって定義してください。
;* • CLOCK 動作させるクロック数の設定
;* (例) 25MHz 動作の場合 CLOCK .EQU 2500
;*
;* • ERASEBLK 2 byte ;対象のアドレスブロックを指定。複数ビット設定可。
;* MSB|<-----|LSB
;* -----
;* EB15|14|13|12|11|10| 9| 8| 7| 6| 5| 4| 3| 2| 1| 0
;* -----
;*
;* • FW_STADR 4 byte ;書き込み先先頭アドレス (FLASH)
;*
;* • FW_EDADR 4 byte ;書き込み先終了アドレス (FLASH)
;*
;* • WKAREA 1 byte ;FLASHメモリ消去/書き込み用として使用するRAMワークエリア
;* ; H'00 -> H'FFEF20 - H'FFFF1F(4KB)
;* ; H'01 -> H'FFDF20 - H'FFEF1F(4KB)
;* ; H'02 -> H'FFCF20 - H'FFDF1F(4KB)
;* ; H'03 -> H'FFBF20 - H'FFCF1F(4KB)
;* ; H'04~H'FF -> 指定不可
;*
;* 4. 正常終了/異常終了時のステータスは下記のとおりとなっています。
;* • ステータス 1 byte ;ROL レジスタ
;* ;H'00 -> 正常終了
;* ;H'01 -> ダウンロードエラー
;* ;H'02 -> 初期化エラー
;* ;H'03 -> 消去エラー
;* ;H'04 -> 書き込みエラー
;*
;* 5. 注意事項

```

```

; *      * ERASEBLK は複数ビット設定可。 *
; *      * WKAREA で指定したアドレスから、FLASH メモリの消去/書き込み用として *
; *      * 4kB のワークエリアを使用。 *
; *      * ポート FWE に関する制御は行っておりません(プルアップ要)。 *
;*****
;*****

        .CPU          300HA:24

        .EXPORT      _FLASH_W          ;
;-----
;   User define
;-----
CLOCK          .EQU    2500              ; 25MHz
ERASEBLK      .EQU    H'FFFF            ; EraseBlock define MSB(EB15)<----->LSB(EB00)
FW_STADR      .EQU    H'00000000        ; Flash writting start address
FW_EDADR      .EQU    H'0007FFFF        ; Flash writting end address
WKAREA        .EQU    H'03              ; Flash control work RAM area
; H'00 -> H'FFEF20 - H'FFFF1F(4KB)
; H'01 -> H'FFDF20 - H'FFEF1F(4KB)
; H'02 -> H'FFCF20 - H'FFDF1F(4KB)
; H'03 -> H'FFBF20 - H'FFCF1F(4KB)
;-----
;   Special function register define
;-----
FCCS          .EQU    H'FEE0B0          ; Flash control/status register
FPCS          .EQU    H'FEE0B1          ; Flash program code select register
FECS          .EQU    H'FEE0B2          ; Flash erase code select register
FKEY          .EQU    H'FEE0B4          ; Flash key code register
FMATS         .EQU    H'FEE0B5          ; Flash mat select register
FTDAR         .EQU    H'FEE0B6          ; Flash transfer destination address register
FVACR         .EQU    H'FEE0B7          ; Flash vector address code control register
FVADRR        .EQU    H'FEE0B8          ; Flash vector address data register(R)
FVADRE        .EQU    H'FEE0B9          ; Flash vector address data register(E)
FVADRH        .EQU    H'FEE0BA          ; Flash vector address data register(H)
FVADRL        .EQU    H'FEE0BB          ; Flash vector address data register(L)

RAMCR         .EQU    H'FEE077          ; RAM control register
;-----
;   Constants define
;-----
RAMTOP        .EQU    H'00FFBF20        ; RAM top address
FLSHWK        .EQU    RAMTOP+((H'03-WKAREA)*H'1000)
; Flash Erase/Write Work address
RUNENTRY      .EQU    FLSHWK+H'10        ; Erase/Write run entry address
INITENTRY     .EQU    FLSHWK+H'20        ; Erase/Write initialize entry address
USERWK        .EQU    FLSHWK+H'800      ; Flash Erase/Write User Program address

COMPLETE      .EQU    H'00              ; COMPLETE(No error)
ERRDWNLD      .EQU    H'01              ; Download error
ERRINIT       .EQU    H'02              ; Initialize error
ERRERASE      .EQU    H'03              ; Erase error
ERRWRITE      .EQU    H'04              ; Write error
    
```

```

        .SECTION      FCODE, CODE, ALIGN=2      ;
;=====
;   Flash erasing/writing control program
;=====
_FLASH_W:
    ORC    #H'80,CCR                ; All interrupt disable
    MOV.L  #FLSHWK+H'1000,SP        ; stack pointer set

    MOV.L  #FCODES,ER5              ; Source    address
    MOV.L  #USERWK,ER6             ; Destination address
    MOV.L  #FCODEE-FCODES,ER4      ; copy size
RAMCOPY:
    MOV.B  @ER5,R0L                ;
    MOV.B  R0L,@ER6                ;
    INC.L  #1,ER5                  ;
    INC.L  #1,ER6                  ;
    DEC.L  #1,ER4                  ;
    BNE   RAMCOPY:8                ;

    MOV.L  #USERWK,ER6             ;
    JMP    @ER6                    ; Jump to program on RAM

FCODES:  .EQU  $                  ;
;=====
;   FLASH ERASE/WRITING PROGRAM
;   RAM にコピーするフラッシュ消去・書き込みプログラム
;=====
_FLASH_PRG_START:
;-----
;   FWE check
;-----
FWELOW:
    MOV.B  @FCCS,R6L              ; FWE@FCCS bit check
    BTST.B  #7,R6L                ;
    BEQ    FWELOW:8               ;
;-----
;   Erase code download
;-----
    MOV.B  #H'01,R6L              ; EPVB@FECS bit set
    MOV.B  R6L,@FECS              ;
    BSR   DOWNLOAD                ; Erase code download
    MOV.B  R0L,R0L                ; R0L(DPFR) check
    BNE   DWNLDERR                ; Not H'00 -> Download error!
;-----
;   Erase initialize
;-----
    BSR   INITIALIZE              ; Erase initialize
    MOV.B  R0L,R0L                ; R0L(FPFR) check
    BNE   INITERR                 ; Not H'00 -> Initialize error!
;-----
;   Erase
;-----
    BSR   ERASE_PROCESS:8         ;
    MOV.B  R0L,R0L                ; R0L(FPFR) check
    
```

```

BNE          ERASEERR:8          ; Not H'00 -> Erase error!
;-----
;      Write code download
;-----
MOV.B #H'01,R6L                  ; PPVS@FPCS bit set
MOV.B R6L,@FPCS                  ;
BSR  DOWNLOAD                    ; Write code download
MOV.B R0L,R0L                    ; R0L(DPFR) check
BNE  DWNLDERR:8                  ; Not H'00 -> Download error!
;-----
;      Write initialize
;-----
BSR  INITIALIZE                  ; Write initialize
MOV.B R0L,R0L                    ; R0L(FPFR) check
BNE  INITERR:8                  ; Not H'00 -> Initialize error!
;-----
;      Write
;-----
BSR  WRITE_PROCESS:8            ;
MOV.B R0L,R0L                    ; R0L(FPFR) check
BNE  WRITEERR:8                 ; Not H'00 -> Write error!
;-----
;      Finish
;-----
MOV.B #COMPLETE,R0L             ; COMPLETE(No error)
RTS                               ;
DWNLDERR:
MOV.B #ERRDWNLD,R0L             ; Download error
RTS                               ;
INITERR:
MOV.B #ERRINIT,R0L              ; Initialize error
RTS                               ;
ERASEERR:
MOV.B #ERRERASE,R0L             ; Erase error
RTS                               ;
WRITEERR:
MOV.B #ERRWRITE,R0L            ; Write error
RTS                               ;
;=====
;      ERASE PROGRAM
;=====
ERASE_PROCESS:
MOV.B #H'5A,R6L                  ; FKEY <- H'5A set(erase enable)
MOV.B R6L,@FKEY                  ;

MOV.B #0,R5L                     ;
MOV.W #ERASEBLK,E5               ;
SUB.L ER0,ER0                    ; FEBS all clear
EP00:
SHLR.W      E5                   ;
BCC  EPNEXT:8                    ; Next block
MOV.B R5L,R0L                    ; R0L(FEBS) set
MOV.L #RUNENTRY,ER6              ; Run entry address
JSR  @ER6                        ; Erase!!
    
```

```

MOV.B R0L,R0L                ; R0L(FPFR) check
BNE EPEND:8                  ; Not H'00 -> Erase error!
EPNEXT:
MOV.W E5,E5                  ;
BEQ EPEND:8                  ; No erase block
INC.B R5L                    ;
BRA EP00:8                   ;
EPEND:
MOV.B #H'00,R6L              ; FKEY <- H'00 set(erase disable)
MOV.B R6L,@FKEY              ;
RTS                          ;
;=====
; WRITE PROGRAM
;=====
WRITE_PROCESS:
MOV.B #H'5A,R6L              ; FKEY <- H'5A set(write enable)
MOV.B R6L,@FKEY              ;
;
MOV.L #FW_STADR,ER5          ;
WP00:
BSR DATAREAD128              ; 128byte data read
MOV.L #_FWRITE_BUFFER,ER0    ; ER0(FMPDR) set
MOV.L ER5,ER1                ; ER1(FMPAR) set
MOV.L #RUNENTRY,ER6          ; Run entry address
JSR @ER6                     ; Write!!
MOV.B R0L,R0L                ; R0L(FPFR) check
BNE WPEND:8                  ; Not H'00 -> Write error!
ADD.L #128,ER5               ;
CMP.L #FW_EDADR,ER5          ;
BLS WP00:8                   ;
WPEND:
MOV.B #H'00,R6L              ; FKEY <- H'00 set(write disable)
MOV.B R6L,@FKEY              ;
RTS                          ;
;=====
; Erase/Write code download routine
;=====
DOWNLOAD:
MOV.B #H'A5,R6L              ; FKEY <- H'A5 set(SCO enable)
MOV.B R6L,@FKEY              ;
;
MOV.B #H'FF,R6L              ; DPFR(@FTDAR) clear
MOV.B R6L,@FLSHWK            ;
MOV.B #WKAREA,R6L           ; FTDAR set
MOV.B R6L,@FTDAR             ;
MOV.L #FCCS,ER6              ; SCO@FCCS bit set
BSET.B #0,@ER6               ;
NOP                           ; Download wait
NOP                           ;
NOP                           ;
NOP                           ;
;
MOV.B #H'00,R6L              ; FKEY <- H'00 set(SCO disable)
MOV.B R6L,@FKEY              ;

```

```

MOV.B @FLSHWK,R0L                ; DPFR(@FTDAR) read
RTS                                ;
;=====
; Erase/Write initialize routine
;=====
INITIALIZE:
;; MOV.W #CLOCK,R0                ; Clock setting
MOV.L #CLOCK,ER0                  ; Clock setting
MOV.L #INITENTRY,ER6             ; Initialize entry address
JSR @ER6                          ; Initialize!!
NOP                               ; wait

RTS                                ;
;=====
; 128bytes DATA read routine
;=====
;-----
; フラッシュメモリへ書き込むデータを、外部デバイスなどから 128 バイト
; ごとに読み込み、_FWRITE_BUFFER へ格納する処理を記述してください。
;-----
DATAREAD128:
RTS

FCODEE: .EQU $                    ;

        .SECTION FRAM,DATA,ALIGN=2
FRAMS:
;*****
_FWRITE_BUFFER .RES.B 128        ;
;*****
FRAME:

.END

```

7. H8/3069F 消去／書き込み用レジスタ・パラメータ一覧

表 7.1 H8/3069F 消去／書き込み用レジスタ

レジスタ名称	略称	R/W	初期値	機能
フラッシュコードコントロールステータス	FCCS	R/W	H'00 H'80	FWE 端子状態のモニタ 消去／書き込み実行中のエラー発生モニタ 内蔵プログラムのダウンロード要求ビット
フラッシュプログラムコードセレクト	FPCS	R/W	H'00	・ダウンロードする書き込みプログラムの選択
フラッシュイレーズコードセレクト	FECS	R/W	H'00	・ダウンロードする消去プログラムの選択
フラッシュキーコード	FKEY	R/W	H'00	H'A5:SCO ビット書き込み許可 H5A:書き込み／消去許可
フラッシュマットセレクト	FMATS	R/W	H'00 H'AA	H'AA:ユーザブートマット選択状態 H'00:ユーザマット選択状態
フラッシュトランスファディスティネーションアドレス	FTDAR	R/W	H'00	・ダウンロード先の内蔵 RAM アドレス指定レジスタ
RAM コントロール	RAMCR	R/W	H'F0	・RAM エミュレーション機能の選択／非選択
フラッシュベクタアドレスコードコントロール	FVACR	R/W	H'00	・消去／書き込み処理、ダウンロード処理中の割り込みベクタテーブル切り替え
フラッシュベクタアドレスコードデータ R	FVADDR	R/W	H'00	・FVACR レジスタによるベクタテーブル空間切り替え時のベクタテーブル先頭アドレス(bit31～bit24)
フラッシュベクタアドレスコードデータ E	FVADRE	R/W	H'00	・FVACR レジスタによるベクタテーブル空間切り替え時のベクタテーブル先頭アドレス(bit23～bit16)
フラッシュベクタアドレスコードデータ H	FVADRH	R/W	H'00	・FVACR レジスタによるベクタテーブル空間切り替え時のベクタテーブル先頭アドレス(bit15～bit8)
フラッシュベクタアドレスコードデータ L	FVADRL	R/W	H'00	・FVACR レジスタによるベクタテーブル空間切り替え時のベクタテーブル先頭アドレス(bit7～bit0)

表 7.2 H8/3069F 消去／書き込み用パラメータ

パラメータ名称	略称	R/W	初期値	割当て	機能
ダウンロードパス・フェイルリザルト	DPFR	R/W	不定	内蔵 RAM	・ダウンロード結果の戻り値
フラッシュパス・フェイルリザルト	FPFR	R/W	不定	R0L	・初期化／消去処理／書き込み処理結果の戻り値
フラッシュプログラムイレース周波数コントロール	FPEFEQ	R/W	不定	ER0	・CPUの動作周波数を設定
フラッシュユーザブランチアドレスセット	FUBRA	R/W	不定	ER1	・ユーザブランチ先のアドレスを設定
フラッシュマルチパーパスアドレスエリア	FMPAR	R/W	不定	ER1	・ユーザマット上の書き込み先の先頭アドレスを設定
フラッシュマルチパーパスデータディスティネーションエリア	FMPDR	R/W	不定	ER0	・ユーザマットに書き込むデータが格納されている領域の先頭アドレスを設定
フラッシュイレースブロックセレクト	FEBS	R/W	不定	ER0	・消去ブロック番号の指定

安全設計に関するお願い

1. 弊社は品質、信頼性の向上に努めておりますが、半導体製品は故障が発生したり、誤動作する場合があります。弊社の半導体製品の故障又は誤動作によって結果として、人身事故、火災事故、社会的損害などを生じさせないような安全性を考慮した冗長設計、延焼対策設計、誤動作防止設計などの安全設計に十分ご注意ください。

本資料ご利用に際しての留意事項

1. 本資料は、お客様が用途に応じた適切なルネサス テクノロジ製品をご購入いただくための参考資料であり、本資料中に記載の技術情報についてルネサス テクノロジが所有する知的財産権その他の権利の実施、使用を許諾するものではありません。
2. 本資料に記載の製品データ、図、表、プログラム、アルゴリズムその他応用回路例の使用に起因する損害、第三者所有の権利に対する侵害に関し、ルネサス テクノロジは責任を負いません。
3. 本資料に記載の製品データ、図、表、プログラム、アルゴリズムその他全ての情報は本資料発行時点のものであり、ルネサス テクノロジは、予告なしに、本資料に記載した製品または仕様を変更することがあります。ルネサス テクノロジ半導体製品のご購入に当たりましては、事前にルネサス テクノロジ、ルネサス販売または特約店へ最新の情報をご確認頂きますとともに、ルネサス テクノロジホームページ(<http://www.renesas.com>)などを通じて公開される情報に常にご注意ください。
4. 本資料に記載した情報は、正確を期すため、慎重に制作したのですが万一本資料の記述誤りに起因する損害がお客様に生じた場合には、ルネサス テクノロジはその責任を負いません。
5. 本資料に記載の製品データ、図、表に示す技術的な内容、プログラム及びアルゴリズムを流用する場合は、技術内容、プログラム、アルゴリズム単位で評価するだけでなく、システム全体で十分に評価し、お客様の責任において適用可否を判断してください。ルネサス テクノロジは、適用可否に対する責任を負いません。
6. 本資料に記載された製品は、人命にかかわるような状況の下で使用される機器あるいはシステムに用いられることを目的として設計、製造されたものではありません。本資料に記載の製品を運輸、移動体用、医療用、航空宇宙用、原子力制御用、海底中継用機器あるいはシステムなど、特殊用途へのご利用をご検討の際には、ルネサス テクノロジ、ルネサス販売または特約店へご照会ください。
7. 本資料の転載、複製については、文書によるルネサス テクノロジの事前の承諾が必要です。
8. 本資料に関し詳細についてのお問い合わせ、その他お気づきの点がございましたらルネサス テクノロジ、ルネサス販売または特約店までご照会ください。